Title
Angular Dependence of Optical Properties of Homogeneous Glasses

Author
Furler, R.A.

Publication Date
1991-07-01

LBL-31727

Lawrence Berkeley Laboratory

UNIVERSITY OF CALIFORNIA

ENERGY & ENVIRONMENT DIVISION

Presented at the ASHRAE 1991 Annual Meeting, Indianapolis, IN, June 22–26, 1991, and to be published in ASHRAE Transactions 1991, Vol. 97, Pt. 2

Angular Dependence of Optical Properties of Homogeneous Glasses

R.A. Furler

June 1991

ENERGY & ENVIRONMENT DIVISION

Prepared for the U.S. Department of Energy under Contract Number DE-AC03-76SF00098

DISCLAIMER

LBL-31727
OM-285

Published in *ASHRAE Transactions* 1991, V.97, Pt. 2

Angular Dependence of Optical Properties of Homogeneous Glasses

R.A. Furler, Ph.D.
Windows and Daylighting Group
Energy & Environment Division
Lawrence Berkeley Laboratory
1 Cyclotron Road
Berkeley, California 94720

June 1991

This work was supported by the Assistant Secretary for Conservation and Renewable Energy, Office of Building Technologies, Building Systems and Materials Division of the U.S. Department of Energy, under Contract No. DE-AC03-76SF00098. Reprinted with permission from ASHRAE (Atlanta, GA), February 25, 1992.

ANGULAR DEPENDENCE OF OPTICAL PROPERTIES OF HOMOGENEOUS GLASSES

R.A. Furler, Ph.D.

ABSTRACT

This paper presents an algorithm to determine the angular dependence of the transmittance and reflectance of homogeneous glazing layers given the reflectance and transmittance at normal incidence, the wavelength, and the thickness. For the commonly used solar and visible properties, "typical" wavelengths are presented that can be used with this procedure to approximate these wavelength-integrated properties. Theoretical calculations have shown that the absolute error of this approximation is less than 1.5% for most clear, low-iron, and absorbing glasses. Existing approximations can yield absolute errors up to 15%. This new approximation can be used with computer programs that calculate energy fluxes through glazing(s) based on the glazing thickness and optical properties at normal incidence.

INTRODUCTION

The purpose of this paper is to provide an algorithm to approximate the angular dependence of the visible and solar transmittance and reflectance of homogeneous glazing layers. The only data required are the visible or solar transmittance and reflectance at normal incidence and the layer thickness. These two data sets are the most commonly used indices supplied by product manufacturers. This algorithm can be used to increase the accuracy of computer programs simulating heat transfer (LBL 1988) and illumination through windows (*Daylighting* 1989).

The angular dependence of 3-mm clear glass is often used to approximate the angular dependence of the transmittance and reflectance for any kind of glass, any thickness, and any number of glazing layers. This can result in errors up to 15%. (Throughout this paper, errors or accuracies given in percent mean the absolute error of transmittance or reflectance, if not noted otherwise.) Even worse, this angular dependence is sometimes not taken into account at all, and the data at normal incidence are used for every angle of incidence. This algorithm achieves an accuracy of better than ±1.5% for the solar properties, which is on the same order or even lower than the consistency of manufactured products. For the visible transmittance and reflectance, this algorithm's accuracy is even better—less than ±0.2%. Generally, the largest errors occur at high angles of incidence.

This paper presents results for homogeneous (i.e., uncoated) glass only. Coated glass is more complex because the multiple reflections within the coating's very thin layers exhibit interference patterns. Furthermore, the optical properties of thin layers, only a few atomic layers thick, are different from the bulk properties. The author hopes to address this topic in the future. The remainder of this report is divided into two sections. In the next section, the formulae are provided with which, for an arbitrary wavelength and thickness and given the complex refractive index, one can accurately determine the transmittance and reflectance of a single homogeneous glazing layer as a function of the angle of incidence (including multiple reflections). Next, given the transmittance and reflectance at normal incidence, the complex refractive index used to approximate the angular properties is determined. The final section describes how the procedure outlined in the following section can be used to approximate angular solar and visible transmittance and reflectance, and the accuracy of the approximated data is evaluated.

OPTICAL PROPERTIES OF HOMOGENEOUS BULK MATERIALS

Optical properties for a homogeneous glazing layer can be directly derived from Fresnel's equations given the complex refractive index (n-ik), the thickness (d), the wavelength (λ), and the angle of incidence (θ). This process is summarized in the first of the following three subsections. In the second subsection, the author explores the possibilities of using one or more normalized universal functions to approximate the angular dependence of transmittance and reflectance for all uncoated glazing materials. The limitations of such a set of universal functions necessitates the development of an algorithm to calculate the refractive index given the optical properties at normal incidence; the refractive index can then be used to calculate transmittances and reflectances at other angles of incidence. This final step is discussed in the third subsection.

Using the Refractive Index to Determine Reflectance and Transmittance

For given real and imaginary parts of the refractive index, n and k, we can determine the reflectance, R, and transmittance, T, including multiple reflections, as a function of the angle of incidence, θ (Figure 1). Using Fresnel's equations and Snell's law (Born and Wolf 1980; Fowles 1975), the relationship between the power reflection coefficient, $r(\theta)$ (or reflectivity), and the real part of the relative refractive index, $n = n_2/n_1$, is derived. For unpolarized radiation incident upon a plane boundary between two optical media (n_1, n_2) at the angle of incidence θ (Figure 1), one finds:

$$r(\theta) = \frac{1}{2}\left(\left(\frac{\sin(\theta-\phi)}{\sin(\theta+\phi)}\right)^2 + \left(\frac{\tan(\theta-\phi)}{\tan(\theta+\phi)}\right)^2\right). \quad (1)$$

The relationship between θ and the angle between the boundary normal and the refracted wave, ϕ (Figure 1), is given by Snell's law:

$$\sin\theta = n\cdot\sin\phi. \quad (2)$$

Reto A. Furler is a Research Associate in the Windows and Daylighting Group, Applied Science Division, Lawrence Berkeley Laboratory, Berkeley, CA.



Figure 1 *Schematic of incident, transmitted, and reflected beams at an interface between two media with the refractive indices* n_1 *and* n_2

At normal incidence, Equation 1 reduces to

$$r_o = \frac{(n-1)^2}{(n+1)^2}. \quad (3)$$

The transmissivity, t, is given by

$$t(\theta) = 1 - r(\theta). \quad (4)$$

The transmittance, $T(\theta)$, and reflectance, $R(\theta)$, counting multiple reflections, are obtained from:

$$T(\theta) = \frac{t(\theta)^2 e^{-\alpha d/\cos\phi}}{1 - r(\theta)^2 e^{-2\alpha d/\cos\phi}} \quad (5a)$$

$$R(\theta) = r(\theta)\left(1 + e^{-\alpha d/\cos\phi}\cdot T(\theta)\right), \quad (5b)$$

where d is the thickness of the plate and k is the imaginary part of the refractive index of medium 2, known as the extinction index, and the absorption coefficient $\alpha = 4\pi k/\lambda$. Equations 1 through 5 determine, for unpolarized light, transmittance T and reflectance R of a homogeneous glass as a function of the angle of incidence θ for known thickness (d), wavelength (λ), and refractive index (n-ik).

Figure 2 graphs reflectance and transmittance vs. the angle of incidence θ for d = 4 mm, λ = 600 nm, n = 1.5, and $k = 5\cdot10^{-6}$. This value of the real part of the refractive index is a good average for silica glasses in the wavelength range from the UV through the NIR. The extinction index chosen represents medium-absorbing glass (i.e., bronze or green). The reflectance and transmittance remain almost constant from normal incidence ($\theta = 0°$) up to about 40°. For grazing incidence ($\theta \approx 90°$), the reflectance is unity and the transmittance equals zero. At this angle, both are independent of n and polarization.

Note that the focus of this paper is to develop a procedure for characterizing the solar optical properties of glazings at specific angles of incidence, θ. Often, hemispherical properties are also of interest. They can be easily calculated from the angular-specific data and the algorithm presented in this paper using the following equations:

$$R_h = \int_0^{\pi/2} R(\theta)\cos\theta\,\sin\theta\,d\theta \quad (6a)$$

$$T_h = \int_0^{\pi/2} T(\theta)\cos\theta\,\sin\theta\,d\theta. \quad (6b)$$

Comparing the Angular Reflectance and Transmittance of Different Glazings

The angular optical properties of uncoated glazings were observed to behave in similar patterns; transmittances and reflectances are roughly constant for angles less than 40°, with transmittances dropping to 0 and reflectances increasing to 1.0 at 90°. This observation raises the question as to whether or not a single function can be used to represent the angular dependence of transmittance and reflectance for different values of n and k (different glazing materials).

Given the functions $T(\theta)$ and $R(\theta)$, we can derive pairs of "normalized" functions—$T_n(\theta)$ for the transmittance and $R_n(\theta)$ for the reflectance—which permit a comparison of the angular dependence of reflectance and transmittance regardless of $T(\theta = 0)$ and $R(\theta = 0)$.

$$T_n(\theta) = \frac{T(\theta)}{T(\theta = 0)} \quad (7)$$



Figure 2 *Calculated transmittance and reflectance of a homogeneous glass pane for given* d, λ, n, *and* k *including multiple reflections vs. the angle of incidence*

$$R_n(\theta) = \frac{R(\theta) - R(\theta = 0)}{1 - R(\theta = 0)}. \qquad (8)$$

The graphs in Figures 3 and 4 show the functions $T_n(\theta)$ and $R_n(\theta)$ for different values of the extinction index ($k = 10^{-4}$, 10^{-5}, 10^{-6}, 10^{-7}, and 10^{-8}; $n = 1.5$; $d = 4$ mm; $\lambda = 600$ nm). For weak absorption ($k \leq 10^{-6}$), we observe only a small dependence of T_n and R_n on k; where absorption is strong ($k \geq 10^{-6}$), T_n and R_n are more sensitive to k. Furthermore, note that k depends very much on λ. However, the dependence on n (in the range $1.4 \leq n \leq 1.6$) and the direct wavelength λ (within 0.3 μm to 4.6 μm) is small.

This dependence of T_n and R_n on k indicates that approximations based on only two "universal" functions, $T_n(\theta)$ and $R_n(\theta)$, for any type of glass might show significant errors, especially for tinted glass exhibiting high values of the extinction index, k. This is demonstrated in Figure 5, which presents the normalized function T_n for the solar transmittance of typical 3-mm clear and 6-mm bronze single glazing and a double glazing consisting of typical 6-mm clear and 6-mm bronze. The normalized transmittance of 3-mm clear glass is often used to approximate the angular transmittance for all glass types and thicknesses. Figure 5 shows this will result in significant errors. Compared to 3-mm clear glass, the other two samples shown in Figure 5 exhibit errors up to 11% and 28% at 70°, with 3% and 4.5% errors at 35°. This can result in very inaccurate calculations of the energy performance of windows, given that most of the time during the day the sun illuminates a window at angles of incidence greater than 45°.

We expect the most exact approximation if we use the specific pair $T_n(\theta)$ or $T(\theta)$ and $R_n(\theta)$ or $R(\theta)$ corresponding to the pair n and k for which $T(0) = T_o$ and $R(0) = R_o$ (Equations 1 through 5, $\theta = 0$). As this would result in an infinite number of "universal" functions, we turn to the development of an algorithm that allows n and k to be determined from R_o and T_o, the reflectance and transmittance at normal incidence of a single glass pane known from measurements. In the following section, this algorithm for homogeneous glasses is described.



Figure 3 *Normalized transmittance of a 4-mm-thick homogeneous glazing layer with the refractive index* n = *1.5 vs. the angle of incidence, calculated for different values of the extinction coefficient* k



Figure 4 *Normalized reflectance of 4-mm-thick homogeneous glazing layer with the refractive index* n = *1.5 vs. the angle of incidence, calculated for different values of the extinction coefficient* k



Figure 5 *Normalized calculated solar transmittance vs. angle of incidence for typical 3-mm clear and typical 6-mm bronze single glazing and a double glazing consisting of typical 6-mm clear and typical 6-mm bronze*

Using the Reflectance and Transmittance to Determine the Refractive Index

We can invert the procedure presented earlier and use the reflectance and transmittance at normal incidence for a given wavelength to determine the refractive index. This is easily done. The real part of the refractive index n is derived by inverting Equation 3

$$n = \frac{1+\sqrt{r_o}}{1-\sqrt{r_o}}. \qquad (9)$$

We use the relationship between $T(\theta)$, $R(\theta)$, and k given in Equation 5b to determine the extinction index, k, considering normal incidence ($\theta = 0$) with $R(0) = R_o$ and $T(0) = T_o$:

$$k = -\frac{\lambda}{4\pi d}\ln\left(\frac{R_o - r_o}{r_o T_o}\right). \qquad (10)$$

We derive the reflectivity at normal incidence, r_o, from Equation 5a using Equation 5b to eliminate the exponential factor in Equation 5a:

$$r_o = \frac{\beta - \sqrt{\beta^2 - 4(2-R_o)R_o}}{2(2-R_o)} \qquad (11)$$

where

$$\beta = T_o^2 - R_o^2 + 2R_o + 1. \qquad (12)$$

Since the reflectance, R_o, and the transmittance, T_o, at normal incidence ($\theta = 0$) are often known for a designated glazing layer (i.e., given thickness), the above procedure can be applied to determine the refractive indices. The refractive indices can then be used to calculate the angular properties (see subsection titled "Using the Refractive Index to Determine Reflectance and Transmittance"). This procedure was used to approximate the angular solar transmittance for 6-mm bronze single glazing. Figure 6 shows a comparison between this approximation and the accurate data. For all angles, the error is less than 0.3%. Compared to the accuracy of measured data of transmittance and reflectance, the error of this approximation is negligible. Since measured data are not absolutely accurate, we require that $0 < R_o + T_o < 1$ when using the above algorithm; otherwise Equations 10 through 12 will result in nonphysical values for k. Even when R_o and T_o satisfy this requirement, the measurement error can cause unreasonable results of k. In both cases, we obtain less accurate approximations of the angular dependence of the transmittance and reflectance (see subsection entitled "Comparing the Angular Reflectance and Transmittance of Different Glazings").

AN APPROXIMATION OF THE ANGULAR DEPENDENCE OF VISIBLE AND SOLAR TRANSMITTANCE AND REFLECTANCE

In the previous section, we derived relations between the optical properties of a single homogeneous glazing layer for a given wavelength. Very often the data available refer to visible or solar radiation at normal incidence. In this section, we adapt the procedure described in the previous section to approximate the angular dependence of these wavelength-integrated properties and define the accuracy of this approximation.



Figure 6 *Comparison between calculated accurate data and the approximation of the angular solar transmittance for typical 6-mm bronze single glazing*

Usually, for a given designation and thickness of glazing, the solar and visible transmittance and reflectance at normal incidence are given by manufacturers in order to describe a glazing performance. These indices have been derived by weighting the spectra for normal incidence with a standard terrestrial solar spectrum (Mecherikunnel 1978) or the sensitivity of a standard human eye (IES 1984), respectively. Thus, they reflect a variety of different values of the wavelength λ and k. Note that k is needed explicitly in Equation 5. It was found that the center wavelengths (area-weighted, not peak average) of the visible (575 nm) and solar spectrum (898 nm) produce effective extinction indices k, which will yield reasonable approximations. With these "effective" wavelengths and the normal solar and visible transmittances and reflectances, we can use Equations 9 through 11 to calculate "effective solar" and "effective visible" n and k values. These, in turn, can be used with the "effective" wavelengths to calculate $T'(\theta)$ and $R'(\theta)$ from Equations 1 through 5, where $T'(\theta)$ and $R'(\theta)$ are approximations of the angular dependence. This procedure is coded in Fortran and is given in the appendix.

To determine the accuracy of our approximations for the visible and solar optical properties of homogeneous glass panes, we first used values of n and k as a function of the wavelength, λ (ranging from .31 μm to 4.6 μm), given by Rubin (1985). For fixed angles of incidence, θ (ranging from 0° to 90°), and glass thickness, d (ranging from 3 mm to 10 mm), we then calculated the spectral reflectance and transmittance for five glass types (clear, bronze, green, grey, low-iron) using Equations 5a and 5b. From these spectra, we determined the wavelength-integrated visible and solar reflectance and transmittance. The data corresponding to normal incidence (0°) were then used to calculate the approximations following the procedure described above (Equations 9 through 12, subsequently Equations 1 through 5).

Comparison of the accurate with the approximated data shows errors of less than 0.2% for the visible and less than 1.5% for the solar properties for all the glass types investigated. If we exclude the "green" glass type, which exhibits the most extreme variations in absorption by wavelength, and consequently in k, the error is even less than 1% for the solar properties. These errors did not vary significantly with thickness in the range between 3 mm and 10 mm but with the angle of incidence, increasing from zero at normal incidence and reaching the maximum at angles between 60° and 85°. For angles between 0° and 35° the error is less than 0.5%.

Furthermore, we evaluated an approximation of the angular dependence of the transmittance and reflectance using one pair of the functions $T_n(\theta)$ and $R_n(\theta)$ corresponding to a single value of n and k for all five glass types mentioned above and for thicknesses between 3 mm and 10 mm. This is a simple extension of the use of a polynomial approximation. For the visible data, the best choice for n and k we could find ($n = 1.525$, $k = 1.2 \cdot 10^{-6}$) results in errors up to $\pm 7.5\%$. The best choice of n and k for the solar data ($n = 1.521$, $k = 2.2 \cdot 10^{-6}$) exhibits errors up to $\pm 6.5\%$. The effective n and k derived from the optical properties of 3-mm clear single glazing do not correspond to this best choice. Therefore, the errors found previously are even higher, as we used the normalized functions of 3-mm clear single glazing as universal functions.

Finally, we investigated the influence of an inaccurate value of the thickness d on the approximation. Usually, a glass thickness declared as 4 mm may vary by a maximum of ± 0.5 mm. If the data supplied for the visible and solar properties correspond to a 3.5-mm-thick sample, but we calculate our approximation based on a thickness of 4 mm, the deviation from the accurate data corresponding to a 3.5-mm sample shows no significant difference compared to the errors mentioned above.

CONCLUSIONS

Existing algorithms can grossly misrepresent the angular solar and visible transmittances and reflectances of common glazing materials. These errors, on the order of 3% to 30%, can lead to proportionately large errors when used in common computer codes that calculate the energy impacts of windows in buildings.

In this paper we present the theory behind, as well as the equations and computer code for, a new algorithm that approximates the angular solar or visible transmittances and reflectances of any uncoated glazing material. The only necessary inputs are the solar or visible properties at normal incidence and the glazing layer thickness, all commonly available properties. The theoretical error for this approximation is less than 1.5%. We are in the process of developing an experimental device to both validate this new algorithm and to allow us to develop similar algorithms for coated glazings.

ACKNOWLEDGMENTS

This work was supported by the Assistant Secretary for Conservation and Renewable Energy, Office of Buildings Technologies, Building Systems and Materials Division of the U.S. Department of Energy under Contract No. DE-AC03-76SF00098.

The author wishes to thank Dariush Arasteh and Michael Rubin of Lawrence Berkeley Laboratory for their valuable comments, suggestions, and support.

REFERENCES

Born, M., and E. Wolf. 1980. *Principles of optics*, 6th ed. New York: Pergammon Press.

Daylighting design tool survey. 1989. Berkeley, CA: Windows and Daylighting Group, Lawrence Berkeley Laboratory.

Fowles, G.R. 1975. *Introduction to modern optics*. New York: Holt, Rinehart and Winston.

IES. 1984. *Lighting handbook*. Reference Volume. New York: Illuminating Engineering Society.

LBL. 1988. WINDOW 3.1: A PC program for analyzing window thermal performance. LBL-25686. Berkeley, CA: Windows and Daylighting Group, Lawrence Berkeley Laboratory.

Mecherikunnel, A.T. 1987. *Proc. seminar on testing solar energy materials and systems*, pp. 83-107. Mt. Prospect, IL: Illuminating Engineering Society.

Rubin, M. 1985. *Solar energy materials*, vol. 12, pp. 275-288.

APPENDIX

```
      SUBROUTINE RT_th_approx(Rtot0,Ttot0,dmm,th,wlnm,  Rtot_th,Ttot_th)

c     R. Furler, LBL, Berkeley, CA 94720  ;  12/2/90


c  I N P U T
c     total reflectance      Rtot0    at (angle of incidence) th=0
c     total transmittance    Ttot0    at (angle of incidence) th=0
c     thickness              dmm      [mm]
c     angle of incidence     th       [deg]
c     wavelength             wlnm     [nm]

c     Rtot0 and Ttot0 can correspond to one single wavelength or represent
c     wavelength integrated reflectance and transmittance. In the latter case
c     use adequate center-wavelength:     575.0nm for VISIBLE
c                                         897.7nm for SOLAR


c  O U T P U T
c     Rtot_th    total Reflectance and Transmittance of a homogeneous
c     Ttot_th    (uncoated) glass plate at the angle th.


c  D E S C R I P T I O N   O F   C A L C U L A T I O N
c     1. calculate reflectivity R and transmissivity T at normal incidence
c     2. determine refractive index (n,k) assuming weak absorption (k<<n)
c     3. calculate R and T at new angle of incidence th for both polarizations
c     4. calculate total reflectance Rtot_th and total transmittance Ttot_th of
c        a thick uncoated glass plate for both polarizations as a function of :
c           -angle of incidence theta (th)
c           -thickness  (dmm)  in mm
c           -refractive index (n,k)
c           -wavelength (wlnm) in nm

c  All reflectance and transmittance data must be in decimal form in every
c  subroutine used!


      real dmm,n,k,wlnm,Rtot0,Ttot0,Rtot_th,Ttot_th,th
      save n,k

      call NKfromRT(Rtot0,Ttot0,dmm,wlnm,  n,k)

      if (k .eq. 1000) then
         print*,'Ttot0=',Ttot0,'   Rtot0=',Rtot0
         print*,'Ttot0 (maybe Rtot0) give unreasonable results for k'
         print*,'Ttot0 is probably to close to zero'
         print*,'enter better Ttot0 and Rtot0'
         RETURN
      end if

      call RTapprox_th(th,dmm,wlnm,n,k,  Rtot_th,Ttot_th)

      RETURN
      END
```

```
      SUBROUTINE NKfromRT(Rtot,Ttot,dmm,wlnm,  n,k)

c     This program calculates the complex refractive index (n,k) from the
c     known (measured) total transmittance Ttot and reflectance Rtot at normal
c     incidence of a homogeneous glass pane.
c     The first step determines the reflectivity R and transmissivity T of one
c     air-glass interface. R and T are then used to determine n and k .
c     No interference assumed.

c     INPUT  : Rtot, Ttot, dmm, wlnm
c     OUTPUT : n, k

c     wavelength  wlnm  in nm
c     thickness   dmm   in mm

      real n,k,num1,num2

      pi=2.*asin(1.)
      d=dmm/1000

      if (Ttot .lt. 0) Ttot=0
      if ((Rtot+Ttot) .gt. 1) then
         RTmax=max(Rtot,Ttot)
         print*, 'Rtot or Ttot have changed from :',Rtot,Ttot
         if (RTmax .eq. Rtot) Rtot=1-Ttot
         if (RTmax .eq. Ttot) Ttot=1-Rtot
         print*, '                               to  :',Rtot,Ttot
      end if

      if (Ttot .gt. 0) then
         num1  = Ttot**2-Rtot**2+2*Rtot+1
         num2  = sqrt(num1**2-4*(2-Rtot)*Rtot)
         deno  = 2*(2-Rtot)
         Rmin  = (num1-num2)/deno
         Rplus = (num1+num2)/deno
c        Rplus >= Rtot ! This is unphysical, therefore Rplus is no solution.
      else
         Rmin=Rtot
      end if

      if (Rmin .lt. 0) then
         print*, 'unphysical result for Rtot, Ttot, wlnm=',Rtot,Ttot,wlnm
      elseif (Rmin .le. Rtot) then
         R=Rmin
         T=1-R
         if (Ttot .ne. 0) then
            n=(1+sqrt(R))/(1-sqrt(R))
            a=(Rtot-R)/R/Ttot
            alpha=-log(a)/2/d
            k=alpha/2/pi*wlnm/1e9
         end if
         if ((Ttot .eq. 0) .or. (k .gt. 1e-2)) k=1000
      end if

      if (k .gt. 1e-4) print*, 'Warning. Out of range of Approximation'

      RETURN
      END
```

```
      SUBROUTINE RTapprox_th(th_deg,dmm,wlnm,n,k,  Rtot_th,Ttot_th)

c     This  program calculates the total reflectance Rtot_th and the total
c     transmittance Ttot_th of a thick (no interference) homogeneous glass
c     plate at the angle of incidence th_deg for unpolarized light.

c     INPUT  : th_deg, dmm, wlnm, n, k
c     OUTPUT : Rtot_th, Ttot_th

c     thickness           dmm    in mm
c     wavelength          wlnm   in nm
c     angle of incidence  th_deg in deg

      real dmm,wl,wlnm,n,k,th_deg

      pi=2.*asin(1.)
      wl=wlnm/1.e9
      th=th_deg/180*pi

      call Rtheta(n,th,  Rp,Rs,Tp,Ts)
      call RTtot_homog(Rp,Tp,n,k,th,dmm/1000,wl,  Rtotp_th,Ttotp_th)
      call RTtot_homog(Rs,Ts,n,k,th,dmm/1000,wl,  Rtots_th,Ttots_th)
      Rtot_th=(Rtots_th+Rtotp_th)/2
      Ttot_th=(Ttots_th+Ttotp_th)/2

      RETURN
      END


      SUBROUTINE Rtheta(n,th,  Rip,Ris,Tip,Tis)

c     This subroutine calculates the reflectivity Rip,Ris and transmissivity
c     Tis,Tip for both TE ('s') and TM ('p') polarization of a single
c     air-glass-interface of a homogeneous glass for a given refractive
c     index (n,k) in function of the angle of incidence th.

c     INPUT  : n, th
c     OUTPUT : Rip, Ris, Tip, Tis

c     thickness           dmm    in mm
c     wavelength          wlnm   in nm
c     angle of incidence  th_deg in deg

      real n

c  Snell's law (for negligible absorptance resp. extinction index) :

      ph=asin(1/n*sin(th))

c  Reflectance- and Transmittance-Intensities (->Fresnel's eq.):

      Ris=((cos(th)-n*cos(ph))/(cos(th)+n*cos(ph)))**2
      Rip=((n*cos(th)-cos(ph))/(n*cos(th)+cos(ph)))**2
      Tis=1-Ris
      Tip=1-Rip

      RETURN
      END
```

```
      SUBROUTINE RTtot_homog(R,T,n,k,th,d,wl,  Rtot,Ttot)

c     this subroutine calculates the total Reflectance Rtot and Transmittance
c     Ttot of a homogeneous glass pane in function of the angle of
c     incidence th. No interference assumed. R and T are the reflectivity
c     and transmissivity, respectively, at this given angle of incidence th.

c     INPUT  : R, T, n, k, th, d, wl
c     OUTPUT : Rtot, Ttot

c     thickness           dmm   in m
c     wavelength          wlnm  in m
c     angle of incidence  th    in rad

      real n,k

      pi=2.*asin(1.)

c  Snell's law (for negligible absorptance resp. extinction index) :

      ph=asin(1/n*sin(th))


c  extenuation within the glass pane for a single path:

      alpha=k*2*pi/wl
      a=exp(-2*alpha*d/cos(ph))


c  Total Reflectance and Transmittance including multiple reflections:

      Ttot=a*T**2/(1-a**2*R**2)
      if (Ttot .le. 0.) Ttot=0.
      Rtot=(1+a*Ttot)*R

      RETURN
      END
```